On Building Trees with Minimum Height, Relationally

Shin-Cheng Mu Computing Laboratory, University of Oxford

scm@comlab.ox.ac.uk

Abstract

The algebraic style of reasoning about programs has been proposed and studied by computing scientists. We rephrase the old problem of building trees of minimum height as an optimisation problem and apply the greedy theorem to derive a linear time algorithm. To put the problem in the right form, we find it necessary to generalise from functions to relations and make use of the converse of a function theorem to write the inverse of a function as a fold.

1 Introduction

Given a list of trees. The task is to combine them into one, retaining the leftto-right order of the trees. How can we combine them so that the height of the resulting tree is as small as possible? The actual contents of the subtrees are not relevant. Therefore we can think of the input as a list of numbers representing the heights of the subtrees. Fig. 1 illustrates one of the best arrangement of subtrees with heights [2, 9, 8, 3, 6, 9], whose height is 11.

Figure 1: A tree with height 11 built from trees with heights [2, 9, 8, 3, 6, 9]

This is actually an instance of the optimal bracketing problem, that is, finding the best way to bracket the expression

a_1 ⊕ a_2 ⊕ a_3 ⊕ . . . a_n

in such a way that the value of the expression is minimal. For this problem, the cost function is a ⊕ b = (max a b) + 1. The optimal bracketing problem is usually solved via a dynamic programming approach, taking cubic time. A linear-time algorithm for this particular problem, however, has been studied in [1].

This paper is about how this problem can be expressed as an optimisation problem of the following form, which has been extensively studied in [2]:

min (⊴) · Λ(foldrn step base)

and how known theories can be applied to derive algorithms to the problem. In the above specification, the function foldrn is the fold for non-empty lists:

foldrn step base [x ] = base x foldrn step base (x : xs) = step x (foldrn step base xs)

Later in this paper, however, we will generalise foldrn step base to be a relation. It folds over the input data to generate an arbitrary solution to the optimisation problem. The Λ operator collects them all. An optimal one with respect to preorder (⊴) is then chosen by min (⊴). These concepts will be explained more precisely in the following sections.

2 Preliminaries

2.1 Relations

During the last decade, the computer scientists gradually discovered the benefit for program derivation to generalise from functions to relations. For our purpose, it suffices to think of a relation f : A ⊏ B as a nondetermistic mapping between the source type A and target type B. For example, the relation split :: List A ⊏ (List A × List A), which splits a list into two sublists, can be defined by

split (xs ++ ys) = (xs, ys)

We write y ← f x to denote that y is a possible value to which x is mapped through relation f . Take split defined above for example, both ([1], [2, 3]) ← split [1, 2, 3] and ([1, 2], [3]) ← split [1, 2, 3] hold.

For a relation f we can find its domain and range:

dom f = {x |∃y : y ∈ B : y ← f x } ran f = {y|∃x : x ∈ A : y ← f x }

Inclusion defines an ordering between relations: f ⊆ g holds if and only if (y ← f x ) ⇒ (y ← g x ). It is the model we choose for program refinement. An specification expr_1 can be refined to expr_2 if expr_2 ⊆ expr_1, while its domain is still preserved (dom expr_1 ⊆ dom expr_2).

Composition of two relations f : B ⊏ C and g : A ⊏ B is defined by z ← (f · g) x ≡ ∃y : y ← g x ∧ z ← f y. The converse of a relation f : A ⊏ B, written f^∘ : B ⊏ A, is defined by x ← f^∘ y ≡ y ← f x . The converse is contravariant, that is, (f · g)^∘ = g^∘ · f^∘.

Throughout this paper, we adopt the convention that f :: A → B denotes a (total) function from A to B, while f :: A ⊏ B denotes a relation.

Let f be a relation of type A ⊏ B. For our purpose it suffices to say that the power transpose operator Λ creates a function Λf :: A → Set B. For a ∈ A, (Λf )a is the set of all values in B which a is mapped to.

(Λf )a = {b|b ← fa}

2.2 Inverting a function

The problem we are dealing with concerns building trees. Functional programmers are aware that flattening a structure is usually performed by a fold operation. Consequently, building a structure is usually performed by the converse operation of unfolding. However, this is not necessarily so. The converse of a function theorem tells us how we can write the inverse of a function as a fold. In this paper we will focus our attention on inverting functions whose range type is a non-empty list, therefore we only need a specialisation of the theorem to non-empty lists. In this paper we denote the type of non-empty lists of As by List^+ A.

Theorem 1 (Converse of a Function Theorem (for non-empty lists)) Let f :: B → List^+ A be given. If base :: A → B and step :: A → B ⊏ B are jointly surjective (ran base ∪ ran step = B) and satisfy

f (base a) = [a] (1) f (step a x ) = a : f x (2)

then f^∘ = foldr step base.

2.3 Minimum

In min (⊴), the symbol (⊴) is an ordering. An ordering is

• reflexive, if x ⊴ x for all x;

• anti-symmetric, if x ⊴ y and y ⊴ x implies x = y;

• transitive, if x ⊴ y and y ⊴ z implies x ⊴ z .

An ordering satisfying all the three conditions above is called a partial order. If it is reflexive and transitive but not necessarily anti-symmetric, it is called a preorder. That means we allow two different solutions to be equally preferred.

A preorder is connected if for all x and y of the correct type, either x ⊴ y or y ⊴ x . That is, every two items can be compared. For a connected preorder (⊴), the relation min (⊴) :: Set A ⊏ A is defined by

x ← min (⊴) xs ≡ x ∈ xs ∧ (∀y : y ∈ xs : y ⊴ x )

It is a relation rather than a function because (⊴) is not necessary anti-symmetric, which means that there may be more than one minimal element in a given set.

Among the many properties of min, we will in particular make use of the one below in the following sections. For a function f , (3) says that if we perform relation f on every item in a set before selecting the minimal one under ordering (⊴), we can also do the selection in the range of f , then perform f afterwards on the selected item only.

f · min (⪯) · Λg ⊆ min (⊴) · Λ(f · g)

where x ⪯ y ≡ (f x ) ⊴ (f y) (3)

2.4 The greedy theorem

The greedy theorem plays the key role in the derivation in this paper.

Theorem 2 (Greedy Theorem (for non-empty lists)) Let base :: A ⊏ B and step :: A → B ⊏ B be relations. If step is monotonic on connected preorder (⊴), on the sense that

y ⊴ x ∧ x^′ ← step a x ⇒ (∃y^′ : y^′ ← step a y : y^′ ⊴ x^′) (4)

then

foldrn (min (⊴) · Λstep) (min (⊴) · Λbase) ⊆ min (⊴) · Λ(foldrn step base)

The expression foldrn step base on the right-hand side generates all the items to be compared by min (⊴) by folding over the input list. The relation base gives us some items to start with, while step takes an item and extends it. The monotonicity condition above in effect means that for two items x and y, y being at least as good as x (with respect to (⊴)), no matter how we extend x to x^′, we can always find a way to extend y to y^′ such that y^′ is not worse than x^′. There is thus no point keeping the worse one, x , in the first place. We need only to keep the best item so far in each stage.

Therefore, we can promote min (⊴) into foldrn. Rather than looking for the minimal one among all the items returned by foldrn, the minimal one is chosen in each step and is the only one passed to the next step.

3 The derivation

3.1 Problem definition

Consider the following datatype definition for tip-valued binary trees

data Tree A = Tip A | Bin (Tree A) (Tree A)

The function computing the height of a tree and flattening a tree can be defined as folds over Tree A in the obvious way:

height :: Tree A → A height = foldTree (⊕) id where a ⊕ b = max a b + 1

flatten :: Tree A → List^+A flatten = foldTree (++) wrap

where wrap a = [a] wraps its argument into a singleton list and foldTree is the fold function for Tree defined by:

foldTree f g (Tip a) = g a foldTree f g (Bin x y) = f (foldTree f g x ) (foldTree f g y) The problem is thus to find, among all the trees which flattens to the given list, the one(s) for which height yields the minimal value.

The inverse of the function flatten, written flatten^∘, relates a list to any tree which flattens to it. It must be a relation because there are in general many trees which could flatten to the same list. The Λ operator enables us to talk about the set of all such trees as a whole. The expression Λ(flatten^∘), having type List^+ A → Set(Tree A), returns the set of all the trees which flatten to the given list. For the purpose of our problem, we choose

x ⊴ y ≡ height x ≤ height y The problem can then be specified as

build = min (⊴) · Λ(flatten^∘)

3.2 Building the tree

As flatten is a fold, flatten^∘ can be a unfold. There is indeed a greedy theorem for problems specified in terms of unfold. In this paper, however, we will make use of the converse of the function theorem and write flatten^∘ as a fold.

An alternative way of representing a tree is the spine representation, in which a tree is represented by the list of subtrees along the left spine, plus the left-most tip. The function roll converts a spine into the ordinary representation, with the help of the Prelude function foldl. It is in fact an isomorphism between Spine A and Tree A.

type Spine A = (A × List(Tree A))

roll :: Spine A → Tree A roll(a, x ) = foldl Bin (Tip a) x The task is to find an inverse for flatten · roll. According to the converse of a function theorem, we need a pair of functions one and add that are jointly surjective and satisfy

f (one a) = [a] f (add a x ) = a : f x

Figure 2: An example of add breaking spine (b, x ) at position p, where x = y ++ z , y = [a_m , a_{m−1}, . . . a_p ] and z = [a_{p−1}, a_{p−2}, . . . a_0]. The s_i ’s are the values on the spine. They are not actually represented in our data structure. After extending x , spine values s_p , s_{p−1}, . . . s_0 are updated.

We claim that the following definition for one and add satisfies the premises.

one a = (a, [ ]) add a (b, x ) = (a, roll (b, y) : z ) where y ++ z = x The condition for one trivially holds. In the definition of add, we use a nondetermistic pattern on the left hand side to break the list x into two parts. Such a matching is always possible because (++) is surjective. To show that add meets the condition, we will need the following fact:

flatten(roll (a, x )) = a : concat(map flatten x ) (5) We reason a : flatten(roll (b, y ++ z )) = {(5)}

a : b : concat(map flatten (y ++ z )) = {concat and map distributes over ++ } a : b : concat(map flatten y) ++ concat(map flatten z ) = {(5), backwards}

a : flatten(roll (b, y)) ++ concat(map flatten z ) = {definition of concat and map} a : concat(map flatten (roll(b, y) : z )) = {(5), backwards}

flatten (roll (a, roll(b, y) : z )) = {definition of add}

flatten (roll (add a (b, y ++ z ))) Thus we have (flatten · roll)^∘ = foldrn add one. Intuitively, we build the spine tree by folding over the non-empty list of values, inserting nodes into the tree one by one. Relation add breaks the spine (b, x ) in an arbitrary position and attaches a to the end, as shown in Fig. 2. There are many ways to break x into y ++ z , which is where the nondeterminism comes from. We will see later that eliminating or reducing this nondeterminism is the key toward deriving an efficient algorithm for optimal bracketing problem.

Having inverted flatten · roll, we can now rephrase our problem definition:

build = {specification}

min (⊴) · Λ(flatten^∘) = {roll isomorphic}

min (⊴) · Λ((flatten · roll · roll^∘)^∘) = {converse contravariant}

min (⊴) · Λ(roll · (flatten · roll)^∘) = {roll function, (3), define x ⪯ y ≡ (roll x ) ⊴ (roll y)} roll · min (⪯) · Λ(flatten · roll)^∘ = {inverting flatten · roll}

roll · min (⪯) · Λ(foldrn add one)

The specification naively generates an exponential number of all possible trees flattening to the same list, and choose an optimal one with respect to preorder (⪯).

3.3 A greedy algorithm

If we can prove that add satisfies the monotonicity condition (4), then we can have a greedy algorithm. However, it is not true with respect to (⪯): a tree with the smallest height does not always remain the best after being extended by add.

Fortunately, add is monotonic on a stronger ordering. If we take (≪) to be the reversed lexicographic ordering on the values along the left spine (that is, first the values on the roots are compared, if they are equal, then the next values on the spines are compared.. and so on), we can prove the monotonic condition with respect to (≪). This choice does make sense: to ensure monotonicity, we need to optimise not only the whole tree, but also all the subtrees on the left spine.

We will explain how we can maintain the monotonicity. For any two spine trees x ≪ y, no matter how x is extended by add, we must find a way to extend y such that the resulting tree is not worse. Suppose the spines of x and y look like in Fig. 3. Note that values on each spines are strictly increasing. Furthermore, by bringing in the context, we can assume that s_m = t_n . Therefore,

1. either the spine values are all the same (i.e. m = n ∧ ∀i : m ≥ i ≥ 0 :

t_i = s_i ), or

Figure 3: Assumption of how x and y look like. Here x = (s_m , [a_{m−1}, a_{m−2}, . . . a_0]) and y = (t_n , [b_{n−1}, b_{n−2}, . . . b_0]). s_{m−1}, s_{m−2}, . . . s_0 and t_{n−1}, t_{n−2}, . . . t_0 are the values on the spine.

Figure 4: How we can extend y when p < r.

2. we can find the first value on x , starting from the root, strictly greater than the corresponding value on y. That is, exists r between m ⊓ n and 0 such that s_r > t_r and ∀i : r > i ≥ 0 : t_i = s_i .

Assume x was extended at position p. If the position comes before the two spines divert from each other, that is, 0 ≤ p ≤ r, we always extend y at the same position p, as in Fig. 4. If p comes after r, that is, p > r, we can always extend y at position r, as in Fig. 5.

Once we know the monotonicity condition holds, we can apply the greedy theorem to refine the specification to

roll · foldrn (min(≪) · Λadd) (min(≪) · Λone)

Figure 5: How we can extend y when p ≥ r.

data Tree a = Tip a | Bin (Tree a) (Tree a)
type SpineC a = (a, [(a,Tree a)])

build :: [Integer] -> Tree Integer
build = rollC . foldrn minadd one

rollC :: SpineC a -> Tree a
rollC (a,x) = foldl join (Tip a) x
  where join x (a,y) = Bin x y

one a = (a,[])

minadd :: Integer -> SpineC Integer -> SpineC Integer
minadd a (b,xs) = (a, minsplit ((b,Tip b):xs))
  where minsplit [x] = [x]
        minsplit (x:y:xs) | a < fst y && fst x < fst y = x:y:xs
                          | otherwise = minsplit (bin x y:xs)
        bin (a,x) (b,y) = (ht a b, Bin x y)

ht a b = (a ‘max‘ b) + 1

Figure 6: Program for Building Trees with Minimum Height

Since one is a function, it is equivalent to

roll · foldrn (min(≪) · Λadd) one

3.4 A further refinement and the code

Further improvements can be made during refining min (≪) · Λadd. We can prove that to find the best position to insert node a on spine x of length n, we do not need to actually check through all the n + 1 possibilities. The minimal result always came from extending x at position p, where p is the maximal index satisfying a ⊕ s_p < s_{p−1}, assuming s_{−1} = ∞. We can start from the left of the spine and choose the first p which satisfies the condition.

In the implementation, we refine the data structure to avoid recomputing the height of each subtree. A spine is represented by type SpineC a = (A, List(A × Tree A)), annotating each subtree along the spine with its height.

The resulting code is shown in Fig. 6, where function build takes a list of integers, denoting tree heights, and returns the tree built out of the list whose height is minimum. Function minadd is the result of the refinement described in this section. It’s not difficult to see that it is a linear time algorithm, since each call to minadd consumes a value, each recursive call to minsplit either returns or joins a node, and each node in the resulting tree is built only once.

4 Conclusion and related work

We have demostrated how the old problem of building trees of minimum height can be rephrased in the new scenario of optimisation problems. To put the problem in the right form, we made use of the converse of a function theorem to express the construction of a tree as a fold on lists. The greedy theorem can then be applied.

At least two reasons justify the generalisation from function to relations. Firstly, it provides us a clean way to express nondeterminism. It is especially convenient when we consider optimisation problem, for which we have multiple choices to make in each step. Without relations, we would have to keep the solutions in lists and mess the specification with bookkeeping details taking care of the lists. Secondly, the inverse of a function is not in general a function. We need to generalise to relations before we can talk about it at all.

The motivation behind this line of research is to extend the result to the optimal bracketing problem in general, for which a non-connected preorder is used and the thinning theorem, a generalisation of the greedy theorem, is applied. The relationship between the thinning theorem and the traditional dynamic programming approach remains a topic for further research.

The greedy theorem, together with a family of theorems useful for optimisation problems has been studied extensively in [2]. The converse of a function theorem and the idea of the spine representation was adopted from [3]. The former has proved to be useful in many applications. The latter, however, is relatively less known. Work is still in progress to find more applications of the converse of a function theorem, as many problems can be specified in terms of the inverse of some known function. The general form of all the three theorems for arbitrary inital datatypes and their proofs can be found in [2].

References

[1] R. Bird. On Building Trees with Minimum Height. Journal of Functional Programming, 7(4):441–445, 1997.

[2] R. Bird and O. de Moor. Algebra of Programming. International Series in Computer Science. Prentice Hall, 1997.

[3] O. de Moor and J. Gibbons. Pointwise Relational Programming. In Algebraic Methodology and Software Technology, May 2000.